--This is your product table with 5 random products
IF OBJECT_ID('tempdb..#Product') IS NOT NULL DROP TABLE #Product
CREATE TABLE #Product
(
	ProductID INT PRIMARY KEY IDENTITY(1,1),
	ProductName VARCHAR(25),
	ProductDescription VARCHAR(max)
)
INSERT INTO #Product (ProductName,ProductDescription) VALUES ('Product Name 1','Product Description 1'),
															 ('Product Name 2','Product Description 2'),
															 ('Product Name 3','Product Description 3'),
															 ('Product Name 4','Product Description 4'),
															 ('Product Name 5','Product Description 5')

--This is your item table.  This would probably just be a truncate statement so that your table is reset for the new values to go in
IF OBJECT_ID ('tempdb..#Item') IS NOT NULL DROP TABLE #Item
CREATE TABLE #Item
(
	ItemID INT PRIMARY KEY IDENTITY(1,1),
	FK_ProductID INT NOT NULL,
	ItemName VARCHAR(25),
	ItemDescription VARCHAR(max)
)

--Declare a bunch of variables for the cursor and insert into the item table process
DECLARE @ProductID INT
DECLARE @ProductName VARCHAR(25)
DECLARE @ProductDescription VARCHAR(max)
DECLARE @RandomItemCount INT
DECLARE @RowEnumerator INT
DECLARE @RandomMaxRecords INT = 10

--We declare a cursor to iterate over the records in product and generate random amounts of items
DECLARE ItemCursor CURSOR
FOR SELECT * FROM #Product

OPEN ItemCursor

FETCH NEXT FROM ItemCursor INTO @ProductID, @ProductName, @ProductDescription
WHILE (@@FETCH_STATUS <> -1)
BEGIN
	
	--Get the Random Number into the variable.  And we only want 1 or more records.  Mod division will produce a 0.
	SELECT @RandomItemCount = ABS(CHECKSUM(NewID())) % @RandomMaxRecords
	SELECT @RandomItemCount = CASE @RandomItemCount WHEN 0 THEN 1 ELSE @RandomItemCount END

	--Iterate on the RowEnumerator to the RandomItemCount and insert item rows
	SET @RowEnumerator = 1
	WHILE (@RowEnumerator <= @RandomItemCount)
	BEGIN
		INSERT INTO #Item (FK_ProductID,ItemName,ItemDescription)
		SELECT @ProductID, REPLACE(@ProductName,'Product','Item'),REPLACE(@ProductDescription,'Product','Item')
		
		SELECT @RowEnumerator = @RowEnumerator + 1
	END
	FETCH NEXT FROM ItemCursor INTO @ProductID, @ProductName, @ProductDescription
END

CLOSE ItemCursor
DEALLOCATE ItemCursor
GO

--Look at the result
SELECT 
	* 
FROM 
	#Product AS P
	RIGHT JOIN #Item AS I ON (P.ProductID = I.FK_ProductID)

--Cleanup
DROP TABLE #Product
DROP TABLE #Item